No. **660**

Symbolic Moment Calculation for the Sojourn Time in M/G/1 Queues with Bernoulli Feedback

by

Hideaki Takagi and Ken-ichi Sakamaki

January 1996

Symbolic Moment Calculation for the Sojourn Time in M/G/1 Queues with Bernoulli Feedback

Hideaki Takagi
Institute of Socio-Economic Planning, University of Tsukuba
1-1-1 Tennoudai, Tsukuba-shi, Ibaraki 305, Japan
and
Ken-ichi Sakamaki
Doctoral Program in Socio-Economic Planning, University of Tsukuba
1-1-1 Tennoudai, Tsukuba-shi, Ibaraki 305, Japan

In queueing systems with feedback of output customers, the time spent by a customer in the system from arrival to final departure (the *sojourn time*) is of a primary interest. The first and second moments of the sojourn time in a first-come first-served M/G/1 queue with Bernoulli feedback were given by Takács [1962]. By using the symbolic formula manipulation functions of *Mathematica*, we can calculate higher-order moments of the sojourn time for Takács's model. We show a *Mathematica* program for this calculation as well as the explicit result for the third moment. The moments of the sojourn time can also be obtained for the exhaustive and gated service systems with multiple server vacations. An attempt shown in this paper suggests a new technique for symbolic computational queueing theory.

Key Words: queue, M/G/1, Bernoulli feedback, sojourn time, server vacation model, symbolic calculation.

1. Introduction

In queueing systems with Bernoulli feedback of output customers, each customer whose service is completed instantaneously joins the tail of the queue with probability $1-\nu$, or leaves the system with probability ν, where $0<\nu\leq 1$. Such systems can be used to model a number of practical systems in which services may be repeated for some reason. Two noteworthy application examples in communication networks are the packet transmission in an error-prone channel or in a contention-based multiple access channel (where ν is the probability of successful transmission), and the segmented message transmission (where the number of segments in a message is geometrically distributed with mean $1/\nu$).

Takács [1963] first studies a first-come first-served (FCFS) M/G/1 system with Bernoulli feedback, and derived the expressions for the mean and the second moment of the sojourn time T of an arbitrary customer. The sojourn time is the time from the arrival to the final departure of a customer, which is one of major performance measures of the system. Further studies on the queue length, the sojourn time, and the waiting time are provided by Disney, McNickle, and Simon [1980], Disney [1981], and Disney, König, and Schmidt [1984]. Takàcs's model has been extended in several ways. Among them are systems with multiple classes of customers. D'avignon and Disney [1977], Simon [1984], and Fontana and Berzosa [1984, 1985] consider priority queues. Takagi [1987], Boxma [1989], de Moraes [1990], Sidi, Levy, and Fuhrmann [1992], and Takine, Takagi, and Hasegawa [1991] investigate cyclic-service queues (polling systems).

The purpose of this paper is to obtain higher-order moments of the customer sojourn time based on Takács's basic model. Since manual calculation is practically impossible due to the complexity in symbolic manipulation of equations, we capitalize on the computer software *Mathematica* for this calculation [Wolfram 1991]. A *Mathematica* program and the explicit expression for the third moment of the sojourn time are shown. We also provide *Mathematica* programs that calculate the moments of the sojourn time in exhaustive and gated service systems with multiple server vacations. These systems are analyzed by Takine, Takagi, and Hasegawa [1991], where only the mean sojourn time are given by manual calculation. We now get higher-order moments of the sojourn time. The higher-order moments can be used to characterize the shape of distribution, such as the coefficients of variation, skewness, and kurtosis.

In a previous work [Takagi and Samamaki 1994], we provided *Mathematica* programs that calculate the moments of the waiting time, sojourn time, queue size, and busy period length, and showed the results for up to the 10th order for an M/G/1 queue without feedback. This method is in contrast with the numerical evaluation of higher-order moments for the length of a busy period by Klimko and Neuts [1973]. Our symbolic moment calculation also complements the use of *Mathematica* for the numerical evaluation of formulas and the discrete event simulation for computer performance analysis shown by Allen [1990, 1994]. All the programs in this paper are written and executed using *Mathematica* Version 2.2 for SPARC from Wolfram Research, Inc. [Wolfram 1991].

2. Notation and Previous Results

Let us introduce the notation for M/G/1 queueing systems with Bernoulli feedback studied in this paper. The customer arrival rate is denoted by λ. The Laplace-Stieltjes transform (LST) of the distribution function (DF), the mean, and the nth moment of the service time are denoted by $B^*(s)$, b, and $b^{(n)}$ $(n = 2, 3, \ldots)$, respectively. The Bernoulli feedback mechanism with parameter ν is described at the beginning of Section 1. The traffic intensity is then given by $\lambda b/\nu$, which is also the server utilization. We assume that $\lambda b < \nu$ for the stability of the system. The LST of the DF for the sojourn time T of a customer is denoted by $T^*(s)$. For systems with multiple server vacations (the vacationing mechanism is described in Section 4), we denote the LST of the DF for the length V of each vacation by $V^*(s)$.

The results for the moments of the sojourn time T that were previously available are summarized as follows. For a system without server vacation, we have [Takács 1963]

$$E[T] = \frac{\lambda b^{(2)} + 2(1 - \lambda b)b}{2(\nu - \lambda b)} \tag{2.1}$$

$$\begin{aligned} E[T^2] &= \frac{\nu^2 - 2\nu}{6(\nu - \lambda b)^2[\nu^2 - \nu(2 + \lambda b) + \lambda b]} \\ &\times \Big\{2\nu[6\lambda b^3 - 6b^2 - 6\lambda b b^{(2)} + 3b^{(2)} + \lambda b^{(3)}] \\ &-12\lambda b^3 + 12b^2 + 6\lambda b b^{(2)} - 2\lambda^2 b b^{(3)} + 3[\lambda b^{(2)}]^2\Big\} \end{aligned} \tag{2.2}$$

For an exhaustive service system with multiple server vacations, we have [Takine, Takagi, Hasegawa 1991]

$$E[T] = \frac{\lambda b^{(2)}}{2(\nu - \lambda b)} + \frac{(1 - \lambda b)b}{\nu - \lambda b} + \frac{E[V^2]}{2E[V]} \tag{2.3}$$

For a gated service system with multiple server vacations, we have [Takine, Takagi, Hasegawa 1991]

$$E[T] = \frac{\lambda b^{(2)}}{2(\nu - \lambda b)} + \frac{(1-\lambda b)b + (1-\nu+\lambda b)E[V]}{\nu - \lambda b} + \frac{E[V^2]}{2E[V]} \tag{2.4}$$

In the present paper, the higher-order moments for T are given for these systems.

3. M/G/1 System without Server Vacations

We first consider an M/G/1 system without server vacations. According to Takács [1963] (see also Takagi [1991, Section 1.6]), the LST $T^*(s)$ of the DF for the sojourn time T of an arbitrary customer is given by

$$T^*(s) = F^*(1, s) \tag{3.1}$$

where $F^*(z,s)$ satisfies the functional equation

$$F^*(z,s) = \nu F_1^*(z,s) + (1-\nu)B^*(s+\lambda-\lambda z)F^*([\nu+(1-\nu)z]B^*(s+\lambda-\lambda z), s) \tag{3.2}$$

Furthermore, $F_1^*(z,s)$ is explicitly given by

$$F_1^*(z,s) = (1-\lambda b/\nu)B^*(s+\lambda-\lambda z) + \Pi^*([\nu+(1-\nu)z]B^*(s+\lambda-\lambda z), s+\lambda-\lambda z) \tag{3.3}$$

where

$$\begin{aligned} \Pi^*(z,s) &:= \sum_{k=1}^{\infty} z^k \int_0^{\infty} e^{-sx} P\{L=k, x < X_+ \le x+dx\} \\ &= \frac{\lambda(1-\lambda b/\nu)z(1-z)[B^*(\lambda-\lambda z) - B^*(s)]}{(s-\lambda+\lambda z)\{[\nu+(1-\nu)z]B^*(\lambda-\lambda z) - z\}}. \end{aligned} \tag{3.4}$$

is the double transform of the joint distribution for the number L of customers in the system and the remaining service time X_+ at an arbitrary time (= at an arrival time) during a busy period.

Therefore it is in principle straightforward to calculate the nth moment

$$E[T^n] = (-1)^n \left[\frac{d^n T^*(s)}{ds^n}\right]_{s=0} = (-1)^n \left[\frac{\partial^n F^*(z,s)}{\partial s^n}\right]_{z=1,s=0} \qquad n = 1, 2, \dots \tag{3.5}$$

of the sojourn time T from (3.2)-(3.4). The procedure is suggested by W. S. Brown in the Appendix of Takács [1963]. For $n = 1$, by differentiating both sides of (3.2) with respect to z and then setting $z = 1$ and $s = 0$, we get

$$\left[\frac{\partial F^*(z,s)}{\partial z}\right]_{z=1,s=0} = \frac{1}{1-(1-\nu)(1-\nu+\lambda b)} \left\{(1-\nu)\lambda b + \nu \left[\frac{\partial F_1^*(z,s)}{\partial z}\right]_{z=1,s=0}\right\} \tag{3.6}$$

By differentiating both sides of (3.2) with respect to s and then setting $z = 1$ and $s = 0$, we get

$$\nu \left[\frac{\partial F^*(z,s)}{\partial s}\right]_{z=1,s=0} = \nu \left[\frac{\partial F_1^*(z,s)}{\partial s}\right] - (1-\nu)b\left\{1 + \left[\frac{\partial F^*(z,s)}{\partial z}\right]_{z=1,s=0}\right\} \tag{3.7}$$

Substituting (3.6) into (3.7), we obtain

$$E[T] = -\left[\frac{\partial F^*(z,s)}{\partial s}\right]_{z=1,s=0}$$

$$= \frac{(1-\nu)(2-\nu)b}{1-(1-\nu)(1-\nu+\lambda b)} + \frac{(1-\nu)b}{1-(1-\nu)(1-\nu+\lambda b)}\left[\frac{\partial F_1^*(z,s)}{\partial z}\right]_{z=1,s=0} \tag{3.8}$$

$$-\left[\frac{\partial F_1^*(z,s)}{\partial s}\right]_{z=1,s=0}$$

where $[\partial F_1^*(z,s)/\partial z]_{z=1,s=0}$ and $[\partial F_1^*(z,s)/\partial s]_{z=1,s=0}$ can be calculated from (3.3) and (3.4). The result is given in (2.1). This result agrees with the evaluation of the mean sojourn time from Little's theorem and the mean queue size in the corresponding batch-arrival M/G/1 queue in which the batch size is geometrically distributed with mean $1/\nu$.

For $n = 2$, the differentiation of both sides of (3.2) and evaluation of the derivatives at $z = 1$ and $s = 0$ yields three equations with respect to

$$\left[\frac{\partial^2 F^*(z,s)}{\partial z^2}\right]_{z=1,s=0}, \quad \left[\frac{\partial^2 F^*(z,s)}{\partial z \partial s}\right]_{z=1,s=0}, \quad \text{and} \quad \left[\frac{\partial^2 F^*(z,s)}{\partial s^2}\right]_{z=1,s=0} \tag{3.9}$$

Solving these equations gives

$$E[T^2] = \left[\frac{\partial^2 F^*(z,s)}{\partial s^2}\right]_{z=1,s=0} \tag{3.10}$$

in terms of the derivatives $[\partial^2 F_1^*(z,s)/\partial z^2]_{z=1,s=0}$, $[\partial^2 F_1^*(z,s)/\partial z \partial s]_{z=1,s=0}$, $[\partial^2 F_1^*(z,s)/\partial s^2]_{z=1,s=0}$, $[\partial F_1^*(z,s)/\partial z]_{z=1,s=0}$, and $[\partial F_1^*(z,s)/\partial s]_{z=1,s=0}$, which are all known from (3.3) and (3.4). We can continue this procedure recursively with respect to n. However, the complexity in symbolic calculation grows rapidly as n increases. Brown spends six pages of Takács [1963] in order to explain the procedure to calculate $E[T^2]$.

We present a *Mathematica* program that calculates $E[T^n]$ automatically in Figure 1. The first and second moment, $E[T]$ and $E[T^2]$, obtained by this program agree with (2.1) and (2.2), respectively. The third moment $E[T^3]$ is shown in the Appendix.

4. M/G/1 Systems with Multiple Server Vacations

We proceed to study exhaustive and gated service M/G/1 systems with multiple server vacations. In an exhaustive service system with multiple server vacations, the server begins a vacation each time the system becomes empty. If the server returns from a vacation to find the system not empty, it starts to work immediately and continues until the system becomes empty again (exhaustive service). If the server returns from a vacation to find no customers waiting, it begins another vacation immediately, and repeats vacations until it finds at least one customer waiting upon returning from a vacation (multiple vacations). The lengths of successive vacations are independent and identically distributed, and also are independent of the arrival and service processes. In a gated service system, when the server returns from a vacation it accepts and serves continuously only those customers that are waiting at that time, deferring the service to all the messages that arrive during the service period until after the next vacation.

We first consider an exhaustive service system. Takine, Takagi, and Hasegawa [1991] show that the LST $T^*(s)$ of the DF for the sojourn time T of an arbitrary customer in this system is again given by (3.1) and (3.2). However, $F_1^*(z,s)$ is now given by

$$\begin{aligned} F_1^*(z,s) &= B^*(s+\lambda-\lambda z)\Omega^*([\nu+(1-\nu)z]B^*(s+\lambda-\lambda z), s+\lambda-\lambda z) \\ &+\Pi^*([\nu+(1-\nu)z]B^*(s+\lambda-\lambda z), s+\lambda-\lambda z) \end{aligned} \tag{4.1}$$

where

$$\Omega^*(z,s) = \frac{(\nu-\lambda b)[V^*(\lambda-\lambda z)-V^*(s)]}{\nu E[V](s-\lambda-\lambda z)} \tag{4.2}$$

$$\Pi^*(z,s) = \frac{(\nu-\lambda b)z[B^*(\lambda-\lambda z)-B^*(s)][1-V^*(\lambda-\lambda z)]}{\nu E[V](s-\lambda+\lambda z)\{[\nu+(1-\nu)z]B^*(\lambda-\lambda z)-z\}} \tag{4.3}$$

A *Mathematica* program that calculates $E[T^n]$ for an exhaustive service system with multiple server vacation is shown in Figure 2. The mean sojourn time $E[T]$ obtained by this program agrees with (2.3). The second moment is given as follows:

$$E[T^2] = \frac{(2-\nu)\nu\{E[V]A + E[V^2]B + E[V^3]C\}}{6\,(\nu-b\,\lambda)^2\,(2\,\nu-\nu^2-b\,\lambda+\nu\,b\,\lambda)\,E[V]} \tag{4.4}$$

where

$$A \equiv 12\,(-1+\nu)\,(-1+b\,\lambda)\,b^2 + 6\,(\nu+b\,\lambda-2\,\nu\,b\,\lambda)\,b^{(2)} + 3\,\lambda^2\,{b^{(2)}}^2 + 2\,\lambda\,(\nu-b\,\lambda)\,b^{(3)}$$

$$B \equiv 3\,(\nu-b\,\lambda)\left(2\,b-2\,b^2\,\lambda+\lambda\,b^{(2)}\right)$$

$$C \equiv 2(-\nu+b\,\lambda)^2 \tag{4.5}$$

For a gated service system, Takine, Takagi, and Hasegawa [1991] show that the LST $T^*(s)$ of the DF for the sojourn time T of an arbitrary customer is given by (3.1), where $F^*(z,s)$ now satisfies the functional equation

$$F^*(z,s) = \nu F_1^*(z,s)+(1-\nu)B^*(s+\lambda-\lambda z)V^*(s+\lambda-\lambda z)F^*([\nu+(1-\nu)z]B^*(s+\lambda-\lambda z), s) \tag{4.6}$$

and

$$\begin{aligned} F_1^*(z,s) &= B^*(s+\lambda-\lambda z)\Omega^*([\nu+(1-\nu)z]B^*(s+\lambda-\lambda z), s+\lambda-\lambda z) \\ &+V^*(s+\lambda-\lambda z)\Pi^*([\nu+(1-\nu)z]B^*(s+\lambda-\lambda z), s+\lambda-\lambda z) \end{aligned} \tag{4.7}$$

with

$$\Omega^*(z,s) = \frac{(\nu-\lambda b)Q([\nu+(1-\nu)z]B^*(\lambda-\lambda))[V^*(\lambda-\lambda z)-V^*(s)]}{\nu E[V](s-\lambda+\lambda z)} \tag{4.8}$$

$$\Pi^*(z,s) = \frac{(\nu-\lambda b)z\{Q([\nu+(1-\nu)z]B^*(\lambda-\lambda z))-Q(z)\}[B^*(\lambda-\lambda z)-B^*(s)]}{\nu E[V](s-\lambda+\lambda z)\{[\nu+(1-\nu)z]B^*(\lambda-\lambda z)-z\}} \tag{4.9}$$

and

$$Q(z) = Q([\nu+(1-\nu)z]B^*(\lambda-\lambda z))V^*(\lambda-\lambda z) \tag{4.10}$$

A *Mathematica* program that calculates $E[T^n]$ for a gated service system with multiple server vacation is shown in Figure 3. The mean sojourn time $E[T]$ obtained by this program agrees with (2.4). The second moment is given as follows:

$$E[T^2] = \frac{(-2+\nu)\{E[V]A + E[V]^2 B + E[V]^3 C + E[V^2]D + E[V^2]E[V]F + E[V^3]G\}}{6\,(-2-b\,\lambda+\nu)\,(-b\,\lambda+\nu)^2\,(b\,\lambda-2\,\nu-b\,\lambda\,\nu+\nu^2)\,E[V]} \tag{4.11}$$

where

$A \equiv \nu\,(-2-b\,\lambda+\nu)\,(12\,b^2\,(-1+b\,\lambda)\,(-1+\nu)+6\,(b\,\lambda+\nu-2\,b\,\lambda\,\nu)\,b^{(2)}+3\,\lambda^2\,{b^{(2)}}^2+2\,\lambda\,(-b\,\lambda+\nu)\,b^{(3)})$,

$B \equiv 12\,b\,(b\,\lambda-4\,\nu-3\,b\,\lambda\,\nu+2\,b^2\,\lambda^2\,\nu+b^3\,\lambda^3\,\nu+6\,\nu^2-2\,b^2\,\lambda^2\,\nu^2-2\,\nu^3+b\,\lambda\,\nu^3)+6\,\lambda\,(b\,\lambda-4\,\nu-4\,b\,\lambda\,\nu-b^2\,\lambda^2\,\nu+4\,\nu^2+2\,b\,\lambda\,\nu^2-\nu^3)\,b^{(2)}$,

$C \equiv 12\,(1+b\,\lambda-\nu)\,(b\,\lambda-2\,\nu-3\,b\,\lambda\,\nu-b^2\,\lambda^2\,\nu+3\,\nu^2+2\,b\,\lambda\,\nu^2-\nu^3)$,

$D \equiv 3\,(b\,\lambda-\nu)\nu\,(-2-b\,\lambda+\nu)\,(-2\,b+2\,b^2\,\lambda-\lambda\,b^{(2)})$,

$F \equiv 6\,(b\,\lambda-\nu)\,(-b\,\lambda+4\,\nu+6\,b\,\lambda\,\nu+2\,b^2\,\lambda^2\,\nu-6\,\nu^2-4\,b\,\lambda\,\nu^2+2\,\nu^3)$,

$G \equiv 2\,\nu\,(-2-b\,\lambda+\nu)\,(-b\,\lambda+\nu)^2$ (4.12)

5. Concluding Remarks

We have shown *Mathematica* programs for the symbolic calculation of the moments for the sojourn time in M/G/1 queues with Bernoulli feedback. These programs only simulate manual calculation, which is rather straightforward. Much efforts are needed after the calculation in order to present the results in a form with simple appearance. It will be of interest to explore the capability of *Mathematica* for the study of queues and other performance evaluation models.

Acknowledgment

This work is supported in part by the Telecommunications Advancement Foundation.

References

[1] Allen, A. O. 1990. *Probability, Statistics, and Queueing Theory with Computer Science Applications.* Second edition, Academic Press, San Diego, California.

[2] Allen, A. O. 1994. *Introduction to Computer Performance Analysis with Mathematica.* Academic Press, Boston.

[3] Boxma, O. J. 1989. Workloads and waiting times in single-server systems with multiple customer classes. *Queueing Systems,* Vol.5, No.1-3 (November), pp.185-214.

[4] D'avignon, G. R., and Disney, R. L. 1977. Queues with instantaneous feedback. *Management Science,* Vol.24, pp.168-180.

[5] de Moraes, L. F. M. 1990. Comments on "Analysis and applications of a multiqueue cyclic service system with feedback." *IEEE Transactions on Communications,* Vol.38, No.2 (February), pp.148-149.

[6] Disney, R. L. 1981. A note on sojourn times in M/G/1 queues with instantaneous Bernoulli feedback. *Naval Research Logistics Quarterly,* Vol.28, pp.679-684.

[7] Disney, R. L., König, D., and Schmidt, V. 1984. Stationary queue-length and waiting-time distributions in single-server feedback queues. *Advances in Applied Probability,* Vol.16, No.2 (June), pp.437-446.

[8] Disney, R. L., McNickle, D. C., and Simon, B. 1980. The M/G/1 queue with instantaneous Bernoulli feedback. *Naval Research Logistics Quarterly,* Vol.27, No.4 (December), pp.635-644.

[9] Fontana, B., and Berzosa, C. D. 1984. Stationary queue-length distribution in an M/G/1 queue with two non-preemptive priorities and general feedback. In *Performance of Computer-Communication Systems,* ed. W. Bux and H. Rudin, pp.333-347, Elsevier, Amsterdam.

[10] Fontana, B., and Berzosa, C. D. 1985. M/G/1 queue with N-priorities and feedback: joint queue-length distributions and response time distribution for any particular sequence. In *Teletraffic Issues in an Advanced Information Society, ITC-11,* ed. M. Akiyama, pp.452-458, Elsevier, Amsterdam.

[11] Klimko, E., and Neuts, M. F. 1973. The single server queue in discrete time - numerical analysis II. *Naval Research Logistics Quarterly,* Vol.20, No.2 (June), pp.305-319.

[12] Sidi, M., Levy, H., and Fuhrmann, S. W. 1992. A queueing network with a single cyclically roving server. *Queueing Systems,* Vol.11, No.1-2 (July), pp.121-144. A corrigengum in "Correction to equation (5.6) in the paper: A queueing network with a single cyclically roving server." *Queueing Systems,* Vol.16, No.1-2, p.193, April 1994.

[13] Simon, B. 1984. Priority queues with feedback. *Journal of Association for Computing Machinery,* Vol.31, pp.134-149.

[14] Takács, L. 1963. A single-server queue with feedback. *The Bell System Technical Journal,* Vol.42, No.2 (March), pp.505-519.

[15] Takagi, H. 1987. Analysis and applications of a multi-queue cyclic service system with feedback. *IEEE Transactions on Communications,* Vol.COM-35, No.2 (February), pp.248-250.

[16] Takagi, H. 1991. *Queueing Analysis : A Foundation of Performance Evaluation, Volume 1: Vacation and Priority Systems, Part 1.* Elsevier Science Publishers B.V., Amsterdam, The Netherlands.

[17] Takagi, H., and Sakamaki, K. 1994. Symbolic moment calculation for an M/G/1 queue. Discussion Paper No. 596, Institute of Socio-Economic Planning, University of Tsukuba, September 1994.

[18] Takine, T., Takagi, H., and Hasegawa, T. 1991. Sojourn times in vacation and polling systems with Bernoulli feedback. *Journal of Applied Probability,* Vol.28, No.2 (June), pp.422-432.

[19] Wolfram, S. 1991. *Mathematica: A System for Doing Mathematics by Computer,* Second edition, Addison-Wesley, Reading, Massachusetts.

```
B[0] = 1   ;   t[0] = 1   ;   B'[0] = -b
Derivative[n_][B][0] = Derivative[n][b]*(-1)^n
T[s] = F[1,s]   ;   F[1,0] = 1
F1[z,s] = (1- lambda b /nu) B[s+ lambda - lambda z] +
   Pai[(nu + (1-nu) z) B [s+ lambda - lambda z] , s+lambda - lambda z]
taylor = Series[F1[z,s],{z,1,3},{s,0,3}]
Pai[z,s]:=( ( lambda ( 1 - lambda b / nu ) z ( 1 - z )
   ( B [ lambda - lambda z ] - B [s] ) ) / ( ( s - lambda
   + lambda z ) ( ( nu + ( 1 - nu ) z ) B [ lambda - lambda z ] - z ) ) )
taylorPi = Series[Pai[z,s],{z,1,3},{s,0,3}]
Derivative[n_,m_][Pai][1,0] := Simplify[Coefficient[Coefficient[taylorPi,s,m],
   (-1+z),n]*n!*m!/. {z->1,s->0}]
Derivative[n_,m_][F1][1,0] := Simplify[Coefficient[Coefficient[taylor,s,m],
   (-1+z),n]*n!*m! /. {z->1,s->0}]
(* the functional equation to be solved *)
eq = F[z,s] == nu F1[z,s] + (1-nu) B[s+ lambda - lambda z] *
   F[(nu + (1-nu) z) B[s+ lambda - lambda z],s]
(* the main part *)
getanswers[0] = {}
equations[n_] := Table[D[D[eq,{s,n-i}],{z,i}] /.{z->1,s->0},{i,0,n}]
answers[n_] := Table[Derivative[n-i,i][F][1,0],{i,0,n}]
getanswers[n_] :=getanswers[n] =  Union[Solve[equations[n],answers[n]][[1]]
   /. getanswers[n-1],getanswers[n-1]]
Mom[n_] := Mom[n] =  Simplify[Derivative[0,n][F][1,0] /. getanswers[n] ]
```

Figure 1. Symbolic calculation of the moments of the sojoun time for an FCFS M/G/1 system with Bernoulli feedback without server vacations.

```
B[0] = 1   ;   V[0] = 1   ;   t[0] = 1   ;   V'[0] = -E[V]
Derivative[n_][V][0] = E[V^n]*(-1)^n
Derivative[n_][B][0] = Derivative[n][b]*(-1)^n
T[s] = F[1,s]   ;   F[1,0] = 1
F1[z,s] = B[s+ lambda - lambda z] *
   Omega[(nu+ (1-nu) z) B[s + lambda - lambda z],  s+lambda - lambda z] +
   Pai[(nu+ (1-nu) z) B [s+ lambda - lambda z] , s+lambda - lambda z]
taylor = Series[F1[z,s],{z,1,3},{s,0,3}]
Pai[z,s] = (nu - lambda b )/( nu E[V]) *
   (z ( B[s] - B[lambda - lambda z])(1 - V[lambda - lambda z]))/
   ((z - (nu + ( 1-nu)z)B[lambda - lambda z])(s- lambda + lambda z))
taylorPi = Series[Pai[z,s],{z,1,3},{s,0,3}]
Omega[z,s] = (nu - lambda b ) / (nu E[V]) *
   (V[lambda - lambda z ] - V[s])/(s - lambda + lambda z)
taylorOmega = Series[Omega[z,s],{z,1,3},{s,0,3}]
Derivative[n_,m_][Pai][1,0] := Simplify[Coefficient[Coefficient[taylorPi,s,m],
   (-1+z),n]*n!*m!/. {z->1,s->0}]
Derivative[n_,m_][Omega][1,0] :=
   Simplify[Coefficient[Coefficient[taylorOmega,s,m],(-1+z),n] * n! * m!
   /. {z->1,s->0} ]
Derivative[n_,m_][F1][1,0] := Simplify[Coefficient[Coefficient[taylor,s,m],
   (-1+z),n]*n!*m! /. {z->1,s->0}]
Pai[1,0] = Coefficient[Coefficient[taylorPi,s,0],(-1+z),0]
Omega[1,0] = Coefficient[Coefficient[taylorOmega,s,0],(-1+z),0]
(* the functional equation to be solved *)
eq = F[z,s] == nu F1[z,s] + (1-nu) B[s+ lambda - lambda z] *
   F[(nu + (1-nu) z) B[s + lambda - lambda z],s]
(* the main part *)
getanswers[0] = {}
equations[n_] := Table[D[D[eq,{s,n-i}],{z,i}] /.{z->1,s->0},{i,0,n}]
answers[n_] := Table[Derivative[n-i,i][F][1,0],{i,0,n}]
getanswers[n_] := getanswers[n] =  Union[Solve[equations[n],answers[n]][[1]]
   /. getanswers[n-1], getanswers[n-1]]
Mom[n_] := Mom[n] =  Simplify[Derivative[0,n][F][1,0] /. getanswers[n]]
```

Figure 2. Symbolic calculation of the moments of the sojoun time for an exhaustive FCFS M/G/1 system with Bernoulli feedback with multiple server vacations.

```
B[0] = 1 ; V[0] = 1 ; t[0] = 1 ; Q[1] = 1 ; B'[0] = -b ; V'[0] = -E[V]
Derivative[n_][V][0] = E[V^n]*(-1)^n
Derivative[n_][B][0] = Derivative[n][b]*(-1)^n
T[s] = F[1,s]    ;    F[1,0] = 1
F1[z,s] = B[s+ lambda - lambda z] *
   Omega[(nu+ (1-nu) z) B[s + lambda - lambda z],  s+lambda - lambda z] +
   Pai[(nu+ (1-nu) z) B [s+ lambda - lambda z] , s+lambda - lambda z] *
   V[s + lambda - lambda z]
taylor = Series[F1[z,s],{z,1,4},{s,0,4}]
Pai[z,s] = (nu - lambda b)/( nu E[V]) *
   (z ( B[s] - B[lambda - lambda z]))/( s - lambda + lambda z) *
   (Q[(nu + (1-nu) z ) B[lambda - lambda z]]-Q[z])/
   (z - ( nu + (1-nu) z) B[lambda - lambda z])
taylorPi = Series[Pai[z,s],{z,1,4},{s,0,4}]
Omega[z,s] = (nu - lambda b) / (nu E[V]) *
   (V[lambda - lambda z ] - V[s])/(s - lambda + lambda z) *
   Q[(nu + (1-nu)z)B[lambda - lambda z]]
taylorOmega = Series[Omega[z,s],{z,1,3},{s,0,3}]
eqq = Q[z] == Q[(nu + (1-nu) z) B[lambda - lambda z]] V[lambda - lambda z]
q1 = Solve[ D[eqq,z] /. z->1 , Derivative[1][Q][1]]
Derivative[1][Q][1] = Derivative[1][Q][1] /. q1 [[1]]
q2 = Solve[ D[eqq,{z,2}] /. z->1 , Derivative[2][Q][1]]
Derivative[2][Q][1] = Derivative[2][Q][1] /. q2 [[1]]
q3 = Solve[ D[eqq,{z,3}] /. z->1 , Derivative[3][Q][1]]
Derivative[3][Q][1] = Derivative[3][Q][1] /. q3 [[1]]
Derivative[n_,m_][Pai][1,0] := Simplify[Coefficient[Coefficient[taylorPi,s,m],
   (-1+z),n]*n!*m!/. {z->1,s->0}]
Derivative[n_,m_][Omega][1,0] :=
Simplify[Coefficient[Coefficient[taylorOmega,s,m],(-1+z),n] * n! * m!
   /. {z->1,s->0} ]
Derivative[n_,m_][F1][1,0] := Simplify[Coefficient[Coefficient[taylor,s,m],
    (-1+z),n]*n!*m! /. {z->1,s->0}]
Pai[1,0] = Coefficient[Coefficient[taylorPi,s,0],(-1+z),0]
Omega[1,0] = Coefficient[Coefficient[taylorOmega,s,0],(-1+z),0]
(* the functional equation to be solved *)
eq = F[z,s] == nu F1[z,s] + (1-nu) B[s+ lambda - lambda z] *
   F[(nu + (1-nu) z) B[s + lambda - lambda z],s] * V[s + lambda - lambda z]
(* the main part *)
getanswers[0] = {}
equations[n_] := Table[D[D[eq,{s,n-i}],{z,i}] /.{z->1,s->0},{i,0,n}]
answers[n_] := Table[Derivative[n-i,i][F][1,0],{i,0,n}]
getanswers[n_] := getanswers[n] =  Union[Solve[equations[n],answers[n]][[1]]
   /. getanswers[n-1],getanswers[n-1]]
Mom[n_] := Mom[n] = Simplify[Derivative[0,n][F][1,0] /. getanswers[n] ]
```

Figure 3. Symbolic calculation of the moments of the sojoun time for a gated FCFS M/G/1 system with Bernoulli feedback with multiple server vacations.

Appendix : Moments of the Sojourn Time in an FCFS M/G/1 System with Bernoulli Feedback without Server Vacations

$$E[T^3] = \frac{B}{A}$$

where

$A \equiv 4\,(b\,\lambda - \nu)^3\,(-b\,\lambda + 2\,\nu + b\,\lambda\,\nu - \nu^2)^2$
$\times\,(-2\,b\,\lambda - b^2\,\lambda^2 + 3\,\nu + 4\,b\,\lambda\,\nu + b^2\,\lambda^2\,\nu - 3\,\nu^2 - 2\,b\,\lambda\,\nu^2 + \nu^3)\,,$

and

$B \equiv -96\,b^5\,\lambda^2\,\nu + 48\,b^6\,\lambda^3\,\nu + 48\,b^7\,\lambda^4\,\nu - 48\,b^4\,\lambda\,\nu^2 + 432\,b^5\,\lambda^2\,\nu^2 - 168\,b^6\,\lambda^3\,\nu^2 - 216\,b^7\,\lambda^4\,\nu^2 +$
$288\,b^3\,\nu^3 + 96\,b^4\,\lambda\,\nu^3 - 1056\,b^5\,\lambda^2\,\nu^3 + 288\,b^6\,\lambda^3\,\nu^3 + 384\,b^7\,\lambda^4\,\nu^3 - 1152\,b^3\,\nu^4 + 168\,b^4\,\lambda\,\nu^4 +$
$1656\,b^5\,\lambda^2\,\nu^4 - 336\,b^6\,\lambda^3\,\nu^4 - 336\,b^7\,\lambda^4\,\nu^4 + 1896\,b^3\,\nu^5 - 720\,b^4\,\lambda\,\nu^5 - 1584\,b^5\,\lambda^2\,\nu^5 + 264\,b^6\,\lambda^3\,\nu^5 +$
$144\,b^7\,\lambda^4\,\nu^5 - 1656\,b^3\,\nu^6 + 936\,b^4\,\lambda\,\nu^6 + 864\,b^5\,\lambda^2\,\nu^6 - 120\,b^6\,\lambda^3\,\nu^6 - 24\,b^7\,\lambda^4\,\nu^6 + 816\,b^3\,\nu^7 -$
$600\,b^4\,\lambda\,\nu^7 - 240\,b^5\,\lambda^2\,\nu^7 + 24\,b^6\,\lambda^3\,\nu^7 - 216\,b^3\,\nu^8 + 192\,b^4\,\lambda\,\nu^8 + 24\,b^5\,\lambda^2\,\nu^8 + 24\,b^3\,\nu^9 - 24\,b^4\,\lambda\,\nu^9$
$+(-48\,b^4\,\lambda^3\,\nu - 24\,b^5\,\lambda^4\,\nu - 120\,b^3\,\lambda^2\,\nu^2 + 228\,b^4\,\lambda^3\,\nu^2 + 168\,b^5\,\lambda^4\,\nu^2 + 96\,b^2\,\lambda\,\nu^3 + 576\,b^3\,\lambda^2\,\nu^3 -$
$468\,b^4\,\lambda^3\,\nu^3 - 396\,b^5\,\lambda^4\,\nu^3 + 288\,b\,\nu^4 - 564\,b^2\,\lambda\,\nu^4 - 1284\,b^3\,\lambda^2\,\nu^4 + 576\,b^4\,\lambda^3\,\nu^4 + 420\,b^5\,\lambda^4\,\nu^4 -$
$864\,b\,\nu^5 + 1236\,b^2\,\lambda\,\nu^5 + 1536\,b^3\,\lambda^2\,\nu^5 - 444\,b^4\,\lambda^3\,\nu^5 - 204\,b^5\,\lambda^4\,\nu^5 + 1032\,b\,\nu^6 - 1380\,b^2\,\lambda\,\nu^6 -$
$984\,b^3\,\lambda^2\,\nu^6 + 192\,b^4\,\lambda^3\,\nu^6 + 36\,b^5\,\lambda^4\,\nu^6 - 624\,b\,\nu^7 + 852\,b^2\,\lambda\,\nu^7 + 312\,b^3\,\lambda^2\,\nu^7 - 36\,b^4\,\lambda^3\,\nu^7 +$
$192\,b\,\nu^8 - 276\,b^2\,\lambda\,\nu^8 - 36\,b^3\,\lambda^2\,\nu^8 - 24\,b\,\nu^9 + 36\,b^2\,\lambda\,\nu^9)\,b^{(2)}$
$+(-30\,b^3\,\lambda^4\,\nu - 6\,b^4\,\lambda^5\,\nu - 12\,b^2\,\lambda^3\,\nu^2 + 42\,b^3\,\lambda^4\,\nu^2 + 30\,b^4\,\lambda^5\,\nu^2 - 30\,b\,\lambda^2\,\nu^3 + 192\,b^2\,\lambda^3\,\nu^3 +$
$36\,b^3\,\lambda^4\,\nu^3 - 48\,b^4\,\lambda^5\,\nu^3 + 180\,\lambda\,\nu^4 + 6\,b\,\lambda^2\,\nu^4 - 474\,b^2\,\lambda^3\,\nu^4 - 102\,b^3\,\lambda^4\,\nu^4 + 30\,b^4\,\lambda^5\,\nu^4 - 450\,\lambda\,\nu^5 +$
$78\,b\,\lambda^2\,\nu^5 + 474\,b^2\,\lambda^3\,\nu^5 + 66\,b^3\,\lambda^4\,\nu^5 - 6\,b^4\,\lambda^5\,\nu^5 + 456\,\lambda\,\nu^6 - 108\,b\,\lambda^2\,\nu^6 - 216\,b^2\,\lambda^3\,\nu^6 - 12\,b^3\,\lambda^4\,\nu^6 -$
$234\,\lambda\,\nu^7 + 60\,b\,\lambda^2\,\nu^7 + 36\,b^2\,\lambda^3\,\nu^7 + 60\,\lambda\,\nu^8 - 12\,b\,\lambda^2\,\nu^8 - 6\,\lambda\,\nu^9)\,[b^{(2)}]^2$
$+(-3\,b^3\,\lambda^6 + 12\,b^2\,\lambda^5\,\nu + 6\,b^3\,\lambda^6\,\nu - 45\,b\,\lambda^4\,\nu^2 - 39\,b^2\,\lambda^5\,\nu^2 - 3\,b^3\,\lambda^6\,\nu^2 + 54\,\lambda^3\,\nu^3 + 132\,b\,\lambda^4\,\nu^3 +$
$51\,b^2\,\lambda^5\,\nu^3 - 117\,\lambda^3\,\nu^4 - 144\,b\,\lambda^4\,\nu^4 - 30\,b^2\,\lambda^5\,\nu^4 + 99\,\lambda^3\,\nu^5 + 69\,b\,\lambda^4\,\nu^5 + 6\,b^2\,\lambda^5\,\nu^5 - 39\,\lambda^3\,\nu^6 -$
$12\,b\,\lambda^4\,\nu^6 + 6\,\lambda^3\,\nu^7)\,[b^{(2)}]^3$
$+(20\,b^4\,\lambda^4\,\nu + 4\,b^5\,\lambda^5\,\nu - 24\,b^3\,\lambda^3\,\nu^2 - 48\,b^4\,\lambda^4\,\nu^2 - 20\,b^5\,\lambda^5\,\nu^2 - 12\,b^2\,\lambda^2\,\nu^3 - 12\,b^3\,\lambda^3\,\nu^3 +$
$56\,b^4\,\lambda^4\,\nu^3 + 32\,b^5\,\lambda^5\,\nu^3 - 32\,b\,\lambda\,\nu^4 + 152\,b^2\,\lambda^2\,\nu^4 + 132\,b^3\,\lambda^3\,\nu^4 - 44\,b^4\,\lambda^4\,\nu^4 - 20\,b^5\,\lambda^5\,\nu^4 + 48\,\nu^5 -$
$332\,b^2\,\lambda^2\,\nu^5 - 160\,b^3\,\lambda^3\,\nu^5 + 20\,b^4\,\lambda^4\,\nu^5 + 4\,b^5\,\lambda^5\,\nu^5 - 96\,\nu^6 + 88\,b\,\lambda\,\nu^6 + 304\,b^2\,\lambda^2\,\nu^6 + 76\,b^3\,\lambda^3\,\nu^6 -$
$4\,b^4\,\lambda^4\,\nu^6 + 76\,\nu^7 - 104\,b\,\lambda\,\nu^7 - 128\,b^2\,\lambda^2\,\nu^7 - 12\,b^3\,\lambda^3\,\nu^7 - 28\,\nu^8 + 48\,b\,\lambda\,\nu^8 + 20\,b^2\,\lambda^2\,\nu^8 + 4\,\nu^9 -$
$8\,b\,\lambda\,\nu^9)\,b^{(3)}$
$+(4\,b^4\,\lambda^6 - 18\,b^3\,\lambda^5\,\nu - 8\,b^4\,\lambda^6\,\nu + 60\,b^2\,\lambda^4\,\nu^2 + 50\,b^3\,\lambda^5\,\nu^2 + 4\,b^4\,\lambda^6\,\nu^2 - 106\,b\,\lambda^3\,\nu^3 - 178\,b^2\,\lambda^4\,\nu^3 -$
$56\,b^3\,\lambda^5\,\nu^3 + 60\,\lambda^2\,\nu^4 + 262\,b\,\lambda^3\,\nu^4 + 200\,b^2\,\lambda^4\,\nu^4 + 30\,b^3\,\lambda^5\,\nu^4 - 126\,\lambda^2\,\nu^5 - 252\,b\,\lambda^3\,\nu^5 - 100\,b^2\,\lambda^4\,\nu^5 -$
$6\,b^3\,\lambda^5\,\nu^5 + 104\,\lambda^2\,\nu^6 + 110\,b\,\lambda^3\,\nu^6 + 18\,b^2\,\lambda^4\,\nu^6 - 40\,\lambda^2\,\nu^7 - 18\,b\,\lambda^3\,\nu^7 + 6\,\lambda^2\,\nu^8)\,b^{(2)}b^{(3)}$
$+(-b^5\,\lambda^6 + 5\,b^4\,\lambda^5\,\nu + 2\,b^5\,\lambda^6\,\nu - 15\,b^3\,\lambda^4\,\nu^2 - 12\,b^4\,\lambda^5\,\nu^2 - b^5\,\lambda^6\,\nu^2 + 31\,b^2\,\lambda^3\,\nu^3 + 42\,b^3\,\lambda^4\,\nu^3 +$
$11\,b^4\,\lambda^5\,\nu^3 - 32\,b\,\lambda^2\,\nu^4 - 80\,b^2\,\lambda^3\,\nu^4 - 45\,b^3\,\lambda^4\,\nu^4 - 5\,b^4\,\lambda^5\,\nu^4 + 12\,\lambda\,\nu^5 + 72\,b\,\lambda^2\,\nu^5 + 80\,b^2\,\lambda^3\,\nu^5 +$
$22\,b^3\,\lambda^4\,\nu^5 + b^4\,\lambda^5\,\nu^5 - 24\,\lambda\,\nu^6 - 64\,b\,\lambda^2\,\nu^6 - 36\,b^2\,\lambda^3\,\nu^6 - 4\,b^3\,\lambda^4\,\nu^6 + 19\,\lambda\,\nu^7 + 26\,b\,\lambda^2\,\nu^7 +$
$6\,b^2\,\lambda^3\,\nu^7 - 7\,\lambda\,\nu^8 - 4\,b\,\lambda^2\,\nu^8 + \lambda\,\nu^9)\,b^{(4)}.$